Liet. matem. rink., T. 42, spec. nr., 312–316

Vieno potencialo teorijos uždavinio tyrimas

Aleksas DOMARKAS (VU, KA), Rimantas-Jonas RAKAUSKAS (KA),
Svajonė VOŠTERIENĖ (KA)
el. paštas: aleksas@ieva.mif.vu.lt, rimantas.rakauskas@tmk.lka.lt, svajone.vosteriene@tmk.lka.lt

1. Įvadas

Erdvėje išdėstytų N sferų ($N \geqslant 2$) potencialas u Dekarto koordinatėse apytiksliai yra išreiškiamas formule

$$u = \sum_{k=1}^{N} \sum_{l=0}^{lmax} \sum_{m=-l}^{l} \frac{b_{k,l,m} \mathrm{CSH}(l, m, r_k)}{|r_k|^{(2l+1)}}, \tag{1}$$

čia $lmax$ yra skleidinio eilė, CSH yra kompleksinės sferinės funkcijos (Complex Solid Harmonic), $r_k = (x - x_k, y - y_k, z - z_k)$, x_k, y_k, z_k – k-osios sferos centro koordinatės, koeficientai $b_{k,l,m}$yra randami iš lygčių sistemos

$$\sum_{\substack{j=1 \\ j \neq i}}^{N} \sum_{l_1=-l}^{l} \sum_{m_1=-l}^{l} b_{j,l_1,m_1} c(l_1, m_1, l, m, i, j) r_{0i}{}^{l} (\varepsilon_i - \varepsilon) l + \frac{b_{i,l,m}(\varepsilon_i l + \varepsilon(l+1))}{r_{0i}{}^{l}}$$
$$= -A_{i,l,m} l r_{0i}{}^{l} (\varepsilon_i - \varepsilon), \tag{2}$$

$i = 1, \ldots, N$, $l = 1, \ldots, lmax$, $m = -l, \ldots, l$, koeficientai $c(l_1, m_1, l, m, i, j)$ yra išreiškiami per kompleksines sferines funkcijas ir Klebšo–Gordano koeficientus, r_{0i} – sferų spinduliai, ε_i, ε – sferų ir aplinkos dielektrinės konstantos, koeficientai charakterizuoja išorinį lauką [1, 2]. Ši sistema tiesinė ir iš viso turi $N(lmax + 1)^2$ lygčių.

Šiame darbe pasitelkiant kompiuterinės algebros sistemą MAPLE 7 randama analizinė sprendinio išraiška. Kompleksines sferines funkcijas mes naudojame iš paketo **orbitals**, kuris yra MAPLE R4 ir MAPLE R5 Share bibliotekose. Klebšo–Gordano koeficientams apskaičiuoti yra sudaryta procedūra $cg(j1, m1, j2, m2, j3, m3)$. Klebšo–Gordano koeficientai yra simboliškai apskaičiuojami atskirai ir saugomi išretintame(sparse) šešiamačiame masyve KG. Tuo yra sutaupomas laikas lygčių sistemos formavimui, nes koeficientus pakanka apskaičiuoti tik vieną kartą. Daugelis Klebšo–Gordano koeficientų yra lygūs nuliui, todėl išretintasis masyvas užima palyginti nedaug vietos. Visi Klebšo–Gordano koeficientai yra saugomi tiksliu simboliniu pavidalu: $r\sqrt{n}$; čia r – racionalusis, n – natūralusis skaičiai. Surastas koeficientų masyvas gali būti panaudojamas kitų uždavinių sprendimui. Mūsų pavyzdyje, kai pirmasis masyvo indeksas kinta nuo -20 iki 20, o likusieji penki – nuo -10 iki 10, 600 MHz kompiuteriu šis masyvas yra apskaičiuojamas per nepilnai tris valandas, o išsaugomas failas cg20.m kompiuterio diskiniame kaupiklyje užima 3015354 baitus.

Bendru atveju gaunama tiesinių lygčių sistema su kompleksiniais koeficientais. Toliau simboliškai sprendžiama ši sistema. Sistemos sprendinį įstatome į formulę (1) ir gauname analizinę potencialo išraišką. Rastą potencialą galima tirti analitiškai ir grafiškai [3]. Šis sprendimo būdas yra ypatingas tuo, kad gautas sprendinys yra tikslus(nes jame nėra apytiksliai išreikštų koeficientų). Toliau galima apskaičiuoti skaitines sprendinio reikšmes atskiruose taškuose su bet kokiu tikslumu. Tai leidžia Maple galimybės. Tiksliai išspręsti lygčių sistemą pavyksta, kai lygčių skaičius yra palyginti nedidelis(iki 200). Esant didesniam lygčių skaičiui tenka pereiti prie apytikslio sprendimo su Maple arba panaudoti MATLAB sistemą [3] arba net panaudoti superkompiuterius su lygiagrečių skaičiavimų galimybėmis.

Toliau yra pateikiamos dvi neprikausomos programos Maple 7 aplinkoje: pirmojoje yra apibrėžiama Klebšo–Gordano koeficientų skaičiavimo procedūra ir formuojamas koeficientų masyvas, antrojoje – formuojama lygčių sistema, simboliškai ji sprendžiama ir randama analizinė potencialo išraiška. Norint šiame darbe sutalpinti analizinę sprendinio išraišką, mes imame dviejų sferų atvejį $N = 2$ ir skleidinio eilę $lmax = 2$. Šiuos parametrus galima keisti, bet labai didinti negalima, nes tiksliam sistemos sprendimui turėtų būti teisinga nelygybė $N(lmax + 1)^2 \leqslant 200$.

2. Klebšo–Gordano koeficientų apskaičiavimas

```
> restart;
> cg:=proc(j1,m1,j2,m2,j3,m3)
> local fa, NUMAX;
> fa:=x->if x<0 then 0 else 1/x! fi;
> NUMAX:=min(j1+j2-j3,j1-m1,j2+m2)+1;
> if abs(j1-j2)>j3
> or j3>j1+j2
> or m3<>m1+m2
> or 2*max(j1,j2,j3)-j1-j2-j3>0
> or  (j1<abs(m1) or j2<abs(m2) or j3<abs(m3))
> then 0 else
> (-1)^(j1-j2+m3)*sqrt((2*j3+1)*(j1+j2-j3)!*
> (j1-j2+j3)!*(-j1+j2+j3)!/(j1+j2+j3+1)!*
> (j1+m1)!*(j1-m1)!*(j2+m2)!*(j2-m2)!*(j3+m3)!*(j3-m3)!)*
> sum('(-1)^(z+j1-j2-m3)*(fa(z)*fa(j1+j2-j3-z)*fa(j1-m1-z)*
> fa(j2+m2-z)*fa(j3-j2+m1+z)*fa(j3-j1-m2+z))','z'=0..NUMAX);
> fi;
> RETURN(%);
> end proc:
> n:=10:
> CG:=array(sparse,-2*n..2*n,-n..n,-n..n,-n..n,-n..n,-n..n):
> for j1 from -2*n to 2*n do
> for j2 from -n to n do
> for j3 from -n to n do
> for m1 from -n to n do
> for m2 from -n to n do
```

```
> for m3 from -n to n do
> cg(j1,j2,j3,m1,m2,m3);
> if %<>0 then CG[j1, j2, j3, m1, m2, m3]:=%; end if;
> od;od;od;od;od;od;
> #save CG, "cg20.m";
```

3. Lygčių sistemos sudarymas

Tarpiniai rezultatai yra neišvedami.

```
> restart;
> lmax:=2:
> with(share):readshare(orbitals,science): with(orbitals):
```

Aplinkos ε:

```
> epsilon:=1:
```

Įvedame sferų skaičių, centrų koordinates, spindulius ir jų ε:

```
> N:=2:
> S1:=[-2,0,0],1,2:
> S2:=[2,0,0],1,2:
> sk:=N*(lmax+1)^2:
> AA:=(i,j,k)->if i=1 and j=0 then 1 else 0 fi:
> for k to N do r||k:=S||k[2];epsilon||k:=S||k[3]; od:
> for i to N do for j  to N do
> rv||i||j:=S||j[1]-S||i[1];r||i||j:=linalg[norm](%,2); od;od:
> eqn:=(i,j)->if j=i then B||i[l,m]*
> (epsilon||i*l+epsilon*(l+1))/(r||i)^l
> else Sum(Sum('B||j[l1,m1]*c(l1,m1,l,m,i,j)*
> (r||i)^l*(epsilon||i-epsilon)
> *l','m1'=-'l1'..'l1'),'l1'=0..'lmax') fi:
> EQs:=seq(sum('eqn(i,j)','j'=1..N)=
> -A(l,m,i)*l*r||i^l*(epsilon||i-epsilon),i=1..N):
> eqs:=[seq(seq(seq(combine(G||k(l,m)),m=-l..l),l=0..lmax),
> k=1..N)]:
> for k from 1 to N do G||k:=unapply(value(EQs[k]),l,m) od:
> A:=AA:
> c:=(l2,m2,l1,m1,i,j)->
> (-1)^(l2+m1-m2)*f2(2*l2+2*l1-1)/f2(2*l2-1)/
> f2(2*l1+1)/r||i||j^(l1+l2+1)*Y(i,j,l1+l2,m2-m1)*
> ((2*l1+2*l2+1)*(2*l2+1)/(2*l1+1))^(1/2)*
> CG(l1+l2,0,l2,0,l1,0)*CG(l1+l2,m1-m2,l2,m2,l1,m1):
> Y:=(i,j,l,m)->expand(subs(r=r||i||j,x=rv||i||j[1],
> y=rv||i||j[2],z=rv||i||j[3],
> 2*sqrt(Pi)*ComplexSolidHarmonic(l,m,x,y,z)/r^l)):
> f2:=n->if type(n,even) then 2^(n/2)*(n/2)!
> elif n=-1 then 1 else n!/f2(n-1) fi:
> CG:=cg:
> ks:=[seq(seq(seq(B||k[l,m],m=-l..l),l=0..lmax),k=1..N)]:
> C:=linalg[genmatrix]([combine(eqs[1])],ks,b||1):
> for k from 2 to sk do
> C:=linalg[stackmatrix](C,linalg[genmatrix]
> ([combine(expand(eqs[k]))],ks,b||k)):
```

```
> end do:
> B:=[seq(b||k[1],k=1..sk)]:
> read "cg20.m";
> cg:=(l2,m2,l1,m1,i,j)->CG[l2,m2,l1,m1,i,j]:
> map(value,C):
> map(combine,%):
> C:=evalm(%):
```

4. Lygčių sistemos sprendimas ir sprendinio formavimas

```
> linalg[linsolve](C,B):
> map(combine,%):
> sol:=linalg[geneqns](linalg[diag](1$sk),
> [seq(seq(seq(B||k[l,m],m=-l..l),l=0..lmax),k=1..N)],%):
> assign(%):
> sum(sum(b[l,m]*CSH(l,m,x,y,z)/r^(2*l+1),m=-l..l),l=0..lmax):
> CSH:=ComplexSolidHarmonic:
> u:=map(normal,%%):
> [seq(subs(seq(seq(b[l,m]=B||k[l,m],m=-l..l),l=0..lmax),u),
> k=1..N)]:
> sum('subs(r=sqrt(x^2+y^2+z^2),x=x-S||k[1][1],
> y=y-S||k[1][2],z=z-S||k[1][3],%[k])','k'=1..N):
> if has(%,I) then evalc(%);subs(I=0,%);map(combine,%,power):fi:
> map(combine,%,power):
```

5. Analizinė potencialo išraiška

Apibrėžiame funkciją $U(x, y, z)$, kuri ir yra ieškomasis potencialas:

```
> U:=unapply(%,x,y,z):U(x,y,z):
> map(simplify,%):
> student[completesquare](%,x);
```

$$-\frac{38272}{307371}\frac{\sqrt{3}z}{\sqrt{\pi}\left((x+2)^2+y^2+z^2\right)^{3/2}}+\frac{128}{307371}\frac{(x+2)\,z\sqrt{3}}{\sqrt{\pi}\left((x+2)^2+y^2+z^2\right)^{5/2}}$$

$$-\frac{38272}{307371}\frac{\sqrt{3}z}{\sqrt{\pi}\left((x-2)^2+y^2+z^2\right)^{3/2}}-\frac{128}{307371}\frac{z\,(x-2)\sqrt{3}}{\sqrt{\pi}\left((x-2)^2+y^2+z^2\right)^{5/2}}$$

```
>
```

Gavome potencialo antrosios eilės ($lmax = 2$) artinio analizinę išraišką. Šiame pavyzdyje vienetinių sferų centrai yra taškuose $O_1(-2,0,0)$ ir $O_2(2,0,0)$. Gautoji funkcija yra trijų kintamųjų. Jos tyrimui galima panaudoti visas kompiuterinės algebros sistemos Maple galimybes.

Literatūra

[1] B.S. Aleksandrov, A.B. Bolotin, N.P. Pošiūnaitė, R.J. Rakauskas, V.K. Šugurov, *Mnogocentrovyje integrali*, VU, Vilnius (1974).

[2] L.D. Landau, E.M. Lifšic, Kvantovaja mechanika, *Teoretičeskaja fizika*, t.III, Nauka, M. (1974).
[3] A. Domarkas, R.J. Rakauskas, S. Cicėnas, Kompiuterinės algebros ir skaitinių metodų sąsaja, *XLII LMD konferencijos darbai*, TEV, Vilnius (2001).

Investigation of one task of the potential theory

A. Domarkas, R.J. Rakauskas, A. Vošterienė

Using computer algebra package Maple we compute aproximative analytical solution. For Clebsch–Gordan coefficients was used sparse matrix array.